



警告

所有安装作业必须由具备资格的
安装工程师执行，
且遵守相关法规及规定。



Hi5a 控制器功能说明书

非接触式工具校准功能





本手册内的信息为 Hyundai Robotics 所有。
未经 Hyundai Robotics 书面授权、不得复制全部或部分内容。
本手册不得提供给第三方、不得用于其它用途。

Hyundai Robotics 保留不经过事先通知而修改本手册的权利。

韩国语印刷 - 2023 年 6 月，第 2 版
Hyundai Robotics Co.、Ltd. 版权所有 © 2023





目录

1. 概要	1-1
1.1. 主要规格	1-3
1.1.1. TCP(Tool Center Point)介绍.....	1-3
1.1.2. 工具校准功能介绍.....	1-4
1.2. 工具校准功能操作顺序.....	1-6
2. 安装	2-1
2.1. 系统要求	2-2
2.1.1. 传感器规格要求.....	2-2
2.2. 系统安装	2-2
2.2.1. 安装传感器.....	2-2
2.2.2. 通信设置.....	2-3
2.2.3. 复制工具校准功能专用机器人程序.....	2-4
2.2.4. 确认字节交换(Byte Swap).....	2-5
2.2.5. 指定用于测量移动的模式.....	2-5
2.2.6. 设置测量移动参数.....	2-7
2.2.7. 设置传感器控制器.....	2-8
3. 功能指南	3-1
3.1. Vector Run	3-2
3.1.1. Vector Run 概要	3-2
3.1.2. Vector run 流程图	3-3
3.2. Reference Run	3-4
3.2.1. Reference Run 概要	3-4
3.2.2. Reference run 2D 流程图	3-5
3.3. Measurement	3-6
3.3.1. Measurement 概要	3-6
3.3.2. Measurement 2D 流程图	3-7
4. 应用实例	4-1
4.1. 作业内容	4-2
4.2. 将工具数据的坐标系方向登记在传感器中	4-3
4.3. 2D 标准数据登记	4-6
4.4. 作业程序示教	4-9
4.5. 确认工具强制变形及作业位置误差	4-10
4.6. 利用传感器测量变形量并执行工具自动校准	4-12
4.7. 工具自动校准完后重新确认作业程序并重启作业	4-15
4.8. 其他作业程序	4-16

目录

图目录

图 1.1 轴原点及工具距离优化功能示教方法	1-3
图 1.2 受损前的示教位置	1-4
图 1.3 因工具受损而偏离的工具位置	1-4
图 1.4 执行工具校准功能	1-5
图 1.5 校准受损工具后恢复原本作业位置	1-5
图 2.1 传感器坐标系	2-2
图 2.2 系统网络结构	2-3
图 2.3 DeviceNet Master 设置及扫描执行界面	2-4
图 2.4 传感器内部的机器人移动模式: Circle	2-5
图 2.5 传感器内部的机器人移动模式: 垂直移动	2-6
图 3.1 Vector Run 功能流程图	3-3
图 3.2 Reference Run 2D 功能流程图	3-5
图 3.3 2D 测量功能流程图	3-7
图 4.1 示例作业的系统构成	4-2
图 4.2 工具变形前工件示教位置	4-9
图 4.3 工具强制变形	4-10
图 4.4 出现误差的状态	4-11
图 4.5 测量工具变形量并执行自动校准	4-14
图 4.6 确认作业程序位置	4-15





1

概要



1. 概要

非接触式工具校准功能

本说明书主要阐述基于 LEONI 传感器的非接触式工具校准功能。

◆【必读说明书】◆

- Hi5a 控制器操作说明书

◆【参考】◆

- 本功能说明书以 Hi5a V40.17-00 版本为准制作。
- 工具校准功能属于可选（选项）功能。先输入“工具数据自动校准(ATDC)”许可证号码后执行相关功能。
- 若要与传感器通信时使用内置 DeviceNet，则需要具备“内置 Fieldbus(DeviceNet)”许可证。



1.1. 主要规格

1.1.1. TCP(Tool Center Point)介绍

机器人为利用工具执行作业，就以 R1 轴法兰为准设置 TCP 的距离及角度。TCP 是机器人为运行而参考的基准点，通过控制各轴按需指定此位置及方向。因此，若用于作业的工具的 TCP 未设置或不正确，则无法确保机器人工作的准确度。

工业机器人为指定这种 TCP 的位置，主要按以下方式指定校准。

1. 如下图所示，用尖头（tip）来固定在外部。
2. 将工具的尖端部分（通常指从焊枪尖端一定突出的焊丝末端）位于外部固定尖头上，并记录其位置。
3. 只更改工具方向，将尖端部分位于外部固定尖头上，并反复执行位置记录。
4. 若都找到轴原点及工具距离则记录 7 点，若只找到工具距离则记录 4 点以上。
5. 在机器人自动校准功能中，执行“轴原点及工具距离优化”，记录 TCP 位置。




图 1.1 轴原点及工具距离优化功能示教方法

6. TCP 的角度可以通过工具数据菜单中的“工具角度校准”功能来进行设置。

1.1.2. 工具校准功能介绍

要保障工业机器人反复作业的质量水平，就需要保持一定的 TCP 位置。若因与周边设备碰撞或反复热变形等原因造成焊枪损坏，会使得 TCP 移动到与原本示教的位置不同的位置，从而会引起相关问题。此时，通过直接修复工具变形（恢复原状），或对 TCP 的变形位置进行校准，可以解决上述问题。然而，修复工具的作业须由工作人员直接用修复专用夹具来执行，从而难以实际执行。因此，通过测量变形量后，按照变形量校准 TCP 的方法更方便，更容易自动化。若使用本公司机器人的工具校准功能，不需要操作人员的介入，即会自动测量机器人的 TCP 校准量，并将其反映在工具数据上，从而能够继续行原本示教的作业。




图 1.2 受损前的示教位置



图 1.3 因工具受损而偏离的工具位置





图 1.4 执行工具校准功能



图 1.5 校准受损工具后恢复原本作业位置

1. 2. 工具校准功能操作顺序





2

安装



2. 安装

非接触式工具校准功能

2.1. 系统要求

2.1.1. 传感器规格要求

- (1) 配备旋转对称形状(就像焊枪一样)的工具;
- (2) 应可将相关工具以圆轨迹、垂直轨迹移动;
- (3) 机器人与工具之间应可进行 DeviceNet 通信。

2.2. 系统安装

2.2.1. 安装传感器

将以下 Light sensor (光传感器) 单元安装在适当的位置。建议安装在机器人工作区域中容易移动且固定后不会更改位置的地方。

将传感器的开口部朝向机器人坐标系的 X 方向或 Y 方向，才能容易执行圆轨迹的工作。



图 2.1 传感器坐标系

2.2.2. 通信设置

- 将控制器 DeviceNet 设置为 Master。
- 将 LEONI 传感器设置为 DeviceNet slave 19 号节点。
- 参考) 传感器的节点号设置方法请参考 LEONI 传感器说明书。
- 如下图所示, 连接 Advintec TCP-Controller。




图 2.2 系统网络结构

- Advintec TCP-Controller 下端的端子如下。其中, 将“FB in”连接到机器人控制器。若有除传感器之外的外部设备, 如焊机等, 则需用“FB out”端口来连接。将“LS”与 Light sensor 连接, 向“PWR”施加 24V。
- 通信线及电源线都连接好后, 设置内置于控制器的 DeviceNet master。根据连接通信板的类型选择相应的 CAN Port。
- 执行“F1: 节点查找”, 确认传感器为 19 号节点。




图 2.3 DeviceNet Master 设置及扫描执行界面

2.2.3. 复制工具校准功能专用机器人程序

将本公司提供的工具校准专用程序文件插入到控制器中。

- G00.IDV: 工具校准程序专用 ID 变量指定文件
- 1000_GETMESVAL.JOB: 将由传感器输入的测量值读取并反映在工具中
- 1001_VECRUN.JOB: Vector run 执行程序
- 1002_REF2D.JOB: 2D Reference run 执行程序
- 1003_REF3D.JOB: 3D Reference run 执行程序
- 1004_REF5D.JOB: 5D Reference run 执行程序
- 1006_MES2D.JOB: 2D Measurement run 执行程序
- 1007_MES3D.JOB: 3D Measurement run 执行程序
- 1008_MES5D.JOB: 5D Measurement run 执行程序
- 1020_INIT_SET.JOB: 设置用于测量移动的参数
- 1021_CALC_C1.JOB: 新建第一个 double-circle 模式
- 1022_CALC_OS.JOB: 新建 oscillate 模式
- 1023_CALC_C2.JOB: 新建第二个 double-circle 模式

2.2.4. 确认字节交换 (Byte Swap)

将由 LEONI 输入的 word 数据的字节交换设为 On。正常设置时，各方向的位置、角度数据输入为 30000。字节交换设为 off 时，计算出的坐标数据会错误输入为 12405。

2.2.5. 指定用于测量移动的模式

用于工具校准的工作由以下两种模式构成。

(1) Double-Circle




图 2.4 传感器内部的机器人移动模式: Circle

将工具从如图所示的起始/结束位置以逆时针方向旋转 2 次。

传感器内部可设置旋转的圆直径。此时，应指定测量移动模式，以使机器人按照相同于设定半径的圆轨迹工作。记录起始点时，圆轨迹中心应垂直于传感器线的交叉点。

(2) Oscillate




图 2.5 传感器内部的机器人移动模式：垂直移动

通过将工具在传感器线交叉点上向上下移动，测量 Z 方向。交叉点为 double-circle 的中心，因此，只有正确示教起始位置，才能测量到正确的 Z 方向。

2.2.6. 设置测量移动参数


用于测量移动的参数可在 1020_INIT_SET.JOB 文件中设置。

giToolNrCorr=1	----→ 输入所要校准的工具编号
giToolNrOrig=25	----→ 输入首次安装的工具编号
'-- moving pattern parameter	
giStPoseDir=0 '0:Y+, 1:X+, 2:Y-, 3:X-	----→ 测量移动起始位置的方向
gdRadius=20	----→ Double circle 模式的半径
gdMaxTrMov=20 'Max trnaslational moving distance	----→ 从起始位置下降的距离
gdOscDist=30 'center - up, down distance	----→ Oscillate 模式上下距离
gdOrientLen=30 'Orientation Length	----→ 2nd double-circle 下降距离
gdMoveSpeed=100 'Moving speed	----→ 测量移动速度



2. 2. 7. 设置传感器控制器

传感器控制器的 MEASURE(测量)与 SETUP(设置)之间转换可以按以下方法进行操作。



要进入各菜单，请按下 OK 键。要移到上级菜单，就需要按下方向键，确认“<-- back - ->”后按 OK 键即可。

设置参数时，长按 OK 键即可保存当前值。若要在未保存设置的状态下退出，请同时长按左右方向键。

使用测量功能前，请务必执行以下设置。除了如下所示的项目之外，其他值可以使用初始值（默认值）。有关传感器设置的详细内容，请参考传感器说明书。

- (1) BASIC - Dimension: 选择 2D、3D、5D。机器人控制器只能执行符合该设置条件的作业程序。
- (2) BASIC - Radius: 设置 Double circle 模式的半径。此结果值须相同于机器人程序的半径。
- (3) BASIC - Z Length: 设置 Oscillate 模式的上下移动距离。此结果值须相同于机器人程序的上下距离。
- (4) BASIC - Orient Length: 为了测量用于 5D 的工具变形角度，设置第二个 Double circle 模式的下降角度。此结果值须相同于机器人程序的移动距离。
- (5) BASIC - Options - Measure direct: 指定测量模式的工具前进方向。基本设置为 down(向下)。
- (6) BASIC - Options - Redundance: 在测量移动上，在传感器内部计算出的值之间的误差范围。若此值过小，根据工具状态的不同，测量移动时会出现错误。默认值为 0.1mm，设置直到测量移动时不会出错的数值。
- (7) USER - Max Tolerance: 与标准工具相比可校准的最大距离。设置为最大值 10.0 mm。
- (8) USER - Max Ori Tol.: 与标准工具相比可校准的最大角度。设置为最大值 10.0 deg。
- (9) GLOBAL - Bus - DevNet - Mac ID: 设置 DeviceNet slave 的节点号。基本上使用（默认值）19 号。若要设置其他编号，则需修改 G00.IDV 文件。若需执行该作业，请询问本公司。
※ 直接修改方法：每 1 号（节点号）修改时，若是单一信号修改 128，若是 Byte 则修改 16，若是 Word 则修改 8。

- (10) GLOBAL - Bus - DevNet - Datarate: 设置 DeviceNet baudrate。请将其设置为与控制器 DeviceNet master 的 baudrate 相同。
- (11) GLOBAL - Bus - DevNet - Byte Swap: 将 word 数据 byte swap 设置为 Yes。







3

功能指南



3. 功能指南

非接触式工具校准功能

3. 1. Vector Run

3. 1. 1. Vector Run 概要

工具校准专用传感器通过利用被测数据计算出 X、Y、Z、RX、RY、RZ 的偏移量(offset)。此时，传感器的基本坐标系与机器人的坐标系不同（不成一致）。要使传感器的坐标系方向与机器人的坐标系成一致，就需要执行 vector run。

Vector run 只要在安装传感器执行一次即可，之后则不需要再次执行。



3.1.2. Vector run 流程图




图 3.1 Vector Run 功能流程图

3. 2. Reference Run

3. 2. 1. Reference Run 概要

工具正常安装好后，执行测量移动并保存标准数据。该标准数据将作为一种比较数据用于工具校准。要新建这种标准数据，就需要执行 Reference run。各标准数据按测量 dimension 设置可分为 2D、3D 和 5D。2D 执行 1 次的 double-circle 测量；3D 执行 1 次的 double-circle、oscillation 测量；5D 执行 1st double-circle、oscillation、2nd double-circle 测量。



3.2.2. Reference run 2D 流程图




图 3.2 Reference Run 2D 功能流程图

3. 3. Measurement

3. 3. 1. Measurement 概要

安装于机器人的工具在与周边设备碰撞，更换工具，持续执行反复作业的过程中，可能慢慢地出现变形。这会影响生产工程，因此，为了防止这种情况的发生，就需要定期进行工具测量。Measurement 在工具校准传感器中执行测量移动并计算工具变形量后，对此进行校准或检测出错误。



3.3.2. Measurement 2D 流程图




图 3.3 2D 测量功能流程图





4

应用实例



4. 应用实例

非接触式工具校准功能

4.1. 作业内容

本章通过简单的作业示例，说明实际使用功能时如何执行校准作业。

各作业构成如下：

- (1) 对沿着台钳边角移动的简单工作进行示教。
- (2) 在正常的工具状态下设置工具校准数据。
- (3) 任意发生工具变形。
- (4) 确认因工具变形而引起的工件误差。
- (5) 执行工具校准功能。
- (6) 通过工件示教位置确认是否适用符合变形工具的校准量。




图 4.1 示例作业的系统构成

4. 2. 将工具数据的坐标系方向登记在传感器中

首先通过使用传感器登记正常的工具数据的工具坐标系。

```
'-- Vector-run extern without signal reflection
'-----
'-----

'-- Initial parameter setting
CALL 1020_INIT_SET

'-- Calculate Circle Pose
CALL 1021_CALC_C1

'-- Activate configuration mode -
gyConfigMode=1
WAIT gxConfModeAct

'-- Activate Vector run extern --
gyVectRunExt=1
WAIT gxVecRunExAct

'-- Set Program Number 1 --
gyPrgNrBit1=1
gyPrgNrBit2=0
gyPrgNrBit3=0
gyPrgNrBit4=0
gyPrgNrBit5=0

'-- Set to Program No.set --
gyPgNrSRIdxSe=1

'-- Wait for confirmation of program number --
WAIT gxPgNrSRIdxRd OR gxError

'-- Vector-run only if input gxPgNrSRIdxRd is set --
IF gxPgNrSRIdxRd=1 THEN

'-- Reset Program Number 1 --
gyPrgNrBit1=0
gyPrgNrBit2=0
gyPrgNrBit3=0
gyPrgNrBit4=0
gyPrgNrBit5=0

gyPgNrSRIdxSe=0

'-- Wait until ready --
WAIT gxReadyForMes
```



```

'-- Reset tool data --
Tool[giToolNrCorr]=Tool[giToolNrOrig]

'-- Double-circle 1 with Tool_corrected --
S1  MOVE L,gpCirPose1,S=50%,A=0,T=giToolNrCorr          '-- Move to starting point
S2  MOVE C,gpCirPose2,S=gdMoveSpeed mm/s,A=3,T=giToolNrCorr  '-- 1st quarter circle
movement
S3  MOVE C,gpCirPose3,S=gdMoveSpeed mm/s,A=3,T=giToolNrCorr  '-- 2nd quarter circle
movement
S4  MOVE C,gpCirPose4,S=gdMoveSpeed mm/s,A=3,T=giToolNrCorr  '-- 3rd quarter circle
movement
S5  MOVE C,gpCirPose1,S=gdMoveSpeed mm/s,A=3,T=giToolNrCorr  '-- 4th quarter circle
movement
S6  MOVE C,gpCirPose2,S=gdMoveSpeed mm/s,A=3,T=giToolNrCorr  '-- 1st quarter circle
movement
S7  MOVE C,gpCirPose3,S=gdMoveSpeed mm/s,A=3,T=giToolNrCorr  '-- 2nd quarter circle
movement
S8  MOVE C,gpCirPose4,S=gdMoveSpeed mm/s,A=3,T=giToolNrCorr  '-- 3rd quarter circle
movement
S9  MOVE C,gpCirPose1,S=gdMoveSpeed mm/s,A=3,T=giToolNrCorr  '-- 4th quarter circle
movement

'-- Execute "Get measured values and Correction"
CALL 1000_GetMesVal

'-- Double-circle 2 with tool_corrected
S10 MOVE L,gpCirPose1,S=50%,A=0,T=giToolNrCorr          '-- Move to starting point
S11 MOVE C,gpCirPose2,S=gdMoveSpeed mm/s,A=3,T=giToolNrCorr  '-- 1st quarter circle
movement
S12 MOVE C,gpCirPose3,S=gdMoveSpeed mm/s,A=3,T=giToolNrCorr  '-- 2nd quarter circle
movement
S13 MOVE C,gpCirPose4,S=gdMoveSpeed mm/s,A=3,T=giToolNrCorr  '-- 3rd quarter circle
movement
S14 MOVE C,gpCirPose1,S=gdMoveSpeed mm/s,A=3,T=giToolNrCorr  '-- 4th quarter circle
movement
S15 MOVE C,gpCirPose2,S=gdMoveSpeed mm/s,A=3,T=giToolNrCorr  '-- 1st quarter circle
movement
S16 MOVE C,gpCirPose3,S=gdMoveSpeed mm/s,A=3,T=giToolNrCorr  '-- 2nd quarter circle
movement
S17 MOVE C,gpCirPose4,S=gdMoveSpeed mm/s,A=3,T=giToolNrCorr  '-- 3rd quarter circle
movement
S18 MOVE C,gpCirPose1,S=gdMoveSpeed mm/s,A=3,T=giToolNrCorr  '-- 4th quarter circle
movement

'-- Execute "Get measured values and Correction"
CALL 1000_GetMesVal

```

4. 应用实例

```
'-- Double-circle 3 with tool_corrected --
S19 MOVE L,gpCirPose1,S=50%,A=0,T=giToolNrCorr           '-- Move to starting point
S20 MOVE C,gpCirPose2,S=gdMoveSpeed mm/s,A=3,T=giToolNrCorr  '-- 1st quarter circle
movement
S21 MOVE C,gpCirPose3,S=gdMoveSpeed mm/s,A=3,T=giToolNrCorr  '-- 2nd quarter circle
movement
S22 MOVE C,gpCirPose4,S=gdMoveSpeed mm/s,A=3,T=giToolNrCorr  '-- 3rd quarter circle
movement
S23 MOVE C,gpCirPose1,S=gdMoveSpeed mm/s,A=3,T=giToolNrCorr  '-- 4th quarter circle
movement
S24 MOVE C,gpCirPose2,S=gdMoveSpeed mm/s,A=3,T=giToolNrCorr  '-- 1st quarter circle
movement
S25 MOVE C,gpCirPose3,S=gdMoveSpeed mm/s,A=3,T=giToolNrCorr  '-- 2nd quarter circle
movement
S26 MOVE C,gpCirPose4,S=gdMoveSpeed mm/s,A=3,T=giToolNrCorr  '-- 3rd quarter circle
movement
S27 MOVE C,gpCirPose1,S=gdMoveSpeed mm/s,A=3,T=giToolNrCorr  '-- 4th quarter circle
movement

'-- Execute "Get measured values and Correction"
CALL 1000_GetMesVal

'-- Acknowledge error during measurement --
'-- Identify and rectify error cause --
IF gxError=1 THEN
'-- Reset tool data --
Tool[giToolNrCorr]=Tool[giToolNrOrig]
gyReset=1
WAIT gxError=0
gyReset=0
ENDIF

ELSE '-- Error when setting program number --
'-- Reset Program Number 1 --
gyPrgNrBit1=0
gyPrgNrBit2=0
gyPrgNrBit3=0
gyPrgNrBit4=0
gyPrgNrBit5=0

gyInitTMeAct=0
ENDIF

'-- Turn off Vector-run extern --
gyVectRunExt=0
WAIT gxVecRunExAct=0

'-- Turn off configuration mode
gyConfigMode=0
WAIT gxConfModeAct=0
END
```

4. 3. 2D 标准数据登记

然后，登记将要在本次用例中使用的 2D 方向标准数据。根据应用范围的不同，还可登记 3D、5D 标准数据。

```
'-- Reference run extern 2D without signal reflection --
'-----
'-----
'-- Initial parameter setting
CALL 1020_INIT_SET

'-- Calculate Circle Pose
CALL 1021_CALC_C1

'-- Activate configuration mode -
gyConfigMode=1
WAIT gxConfModeAct

'-- Activate reference run extern --
gyRefRunExt=1
WAIT gxRefRunExAct=1

'-- Set Program Number 1 --
gyPrgNrBit1=1
gyPrgNrBit2=0
gyPrgNrBit3=0
gyPrgNrBit4=0
gyPrgNrBit5=0

'-- Set referencing for tool --
gyConfigTool=1

'-- Set referencing for diameter --
gyRefDiameter=1

'-- Set to Program No.set --
gyPgNrSRIdxSe=1

'-- Wait for confirmation of program number --
WAIT gxPgNrSRIdxRd OR gxError
'-- Reference run only when input gxPgNrSRIdxRd is set --
IF (gxPgNrSRIdxRd=1) THEN
'-- Reset Program Number 1 --
gyPrgNrBit1=0
gyPrgNrBit2=0
gyPrgNrBit3=0
gyPrgNrBit4=0
gyPrgNrBit5=0
```



```

'-- Reset referencing for tool --
gyConfigTool=0
'-- Reset referencing for diameter --
gyRefDiameter=0
'-- Reset to program number set --
gyPgNrSRIdxSe=0
'-- Wait until ready --
WAIT gxReadyForMes
'-- Reset tool data --
Tool[giToolNrCorr]=Tool[giToolNrOrig]
'-- Double-circle with tool_corrected
S1  MOVE L,gpCirPose1,S=50%,A=0,T=giToolNrCorr      '-- Move to starting point
S2  MOVE C,gpCirPose2,S=50mm/s,A=3,T=giToolNrCorr   '-- 1st quarter circle movement --
S3  MOVE C,gpCirPose3,S=50mm/s,A=3,T=giToolNrCorr   '-- 2nd quarter circle movement --
S4  MOVE C,gpCirPose4,S=50mm/s,A=3,T=giToolNrCorr   '-- 3rd quarter circle movement --
S5  MOVE C,gpCirPose1,S=50mm/s,A=3,T=giToolNrCorr   '-- 4th quarter circle movement --
S6  MOVE C,gpCirPose2,S=50mm/s,A=3,T=giToolNrCorr   '-- 1st quarter circle movement --
S7  MOVE C,gpCirPose3,S=50mm/s,A=3,T=giToolNrCorr   '-- 2nd quarter circle movement --
S8  MOVE C,gpCirPose4,S=50mm/s,A=3,T=giToolNrCorr   '-- 3rd quarter circle movement --
S9  MOVE C,gpCirPose1,S=50mm/s,A=0,T=giToolNrCorr   '-- 4th quarter circle movement --

'-- Execute "Get measured values and Correction"
CALL 1000_GetMesVal

'-- Acknowledge error during measurement --
'-- Identify and rectify error cause --
IF gxError=1 THEN
'-- Reset tool data --
Tool[giToolNrCorr]=Tool[giToolNrOrig]
gyReset=1
WAIT gxError=0
gyReset=0
ENDIF
ELSE '-- Error when setting program number --

'-- Reset Program Number 1 --
gyPrgNrBit1=0
gyPrgNrBit2=0
gyPrgNrBit3=0
gyPrgNrBit4=0
gyPrgNrBit5=0

```



```
'-- Reset referencing for tool --
gyConfigTool=0
'-- Reset referencing for diameter -
gyRefDiameter=0
'-- Reset to program no.set --
gyPgNrSRIdxSe=0
ENDIF
'-- Turn off reference run extern --
gyRefRunExt=0
WAIT gxRefRunExAct=0
'-- Turn off configuration mode --
gyConfigMode=0
WAIT gxConfModeAct=0
END
```



4.4. 作业程序示教

沿着如下图所示的方形工件边角，示教作业程序。

==== 0901. JOB ===

’确认程序

```
S1    MOVE P, S=50%, A=3, T=1  
S2    MOVE L, S=30%, A=0, T=1  
      DELAY 0.1  
S3    MOVE L, S=40mm/s, A=3, T=1      DELAY 0.1  
S4    MOVE L, S=40mm/s, A=3, T=1      DELAY 0.1  
S5    MOVE P, S=50%, A=3, T=1  
      END
```

=====




图 4.2 工具变形前工件示教位置

4.5. 确认工具强制变形及作业位置误差

为了执行测试，使机器人的工具强制变形。



图 4.3 工具强制变形

根据工具变形的程度，确认工件出现的误差。



图 4.4 出现误差的状态

4. 6. 利用传感器测量变形量并执行工具自动校准

```

'-- Measurement 2D without signal reflection
'-- Initial parameter setting
CALL 1020_INIT_SET

'-- Calculate Circle Pose
CALL 1021_CALC_C1

'-- Reset pending error
IF gxError=1 THEN
gyReset=1
WAIT gxError=0
gyReset=0
ENDIF

'-- Set Program Number 1 --
gyPrgNrBit1=1
gyPrgNrBit2=0
gyPrgNrBit3=0
gyPrgNrBit4=0
gyPrgNrBit5=0

'-- WAIT ready --
WAIT gxReadyForMes

'-- Reset tool data --
Tool[giToolNrCorr]=Tool[giToolNrOrig]
'-- Double-circle with Tool[giToolNrCorr]
S1  MOVE L, gpCirPose1, S=50%, A=0, T=giToolNrCorr      '-- Move to starting point      -
S2  MOVE C, gpCirPose2, S=50mm/s, A=3, T=giToolNrCorr   '-- 1st quarter circle movement --
S3  MOVE C, gpCirPose3, S=50mm/s, A=3, T=giToolNrCorr   '-- 2nd quarter circle movement --
S4  MOVE C, gpCirPose4, S=50mm/s, A=3, T=giToolNrCorr   '-- 3rd quarter circle movement --
S5  MOVE C, gpCirPose1, S=50mm/s, A=3, T=giToolNrCorr   '-- 4th quarter circle movement --
S6  MOVE C, gpCirPose2, S=50mm/s, A=3, T=giToolNrCorr   '-- 1st quarter circle movement --
S7  MOVE C, gpCirPose3, S=50mm/s, A=3, T=giToolNrCorr   '-- 2nd quarter circle movement --
S8  MOVE C, gpCirPose4, S=50mm/s, A=3, T=giToolNrCorr   '-- 3rd quarter circle movement --
S9  MOVE C, gpCirPose1, S=50mm/s, A=3, T=giToolNrCorr   '-- 4th quarter circle movement --

```



```
'-- Execute "Get measured values and Correction"
'-- Get measurement result and Correction --
'-----
V100%=19 'sensor node number
'-- Set measurement movement to finished --
gyMeMovFinish=1

'-- Wait until measurement result is calculated or error --
WAIT gxMesResCalc=1 OR gxError=1

'-- Read measured values only if gxMesResCalc == 1 --
IF gxMesResCalc=1 THEN
gdCorrX =(gxXOffsetBit - 30000) / 100.0 '-- X-Offset --
gdCorrY =(gxYOffsetBit - 30000) / 100.0 '-- Y-Offset --
gdCorrZ =(gxZOffsetBit - 30000) / 100.0 '-- Z-Offset --
gdCorrRX=(gxX OriOffset - 30000) / 100.0 '-- Rotation around X --
gdCorrRY=(gxY OriOffset - 30000) / 100.0 '-- Rotation around Y --
gdCorrRZ=(gxZ OriOffset - 30000) / 100.0 '-- Rotation around Y --
ENDIF

'-- If error gxError set correction to zero
IF gxError=1 THEN
gdCorrX =0
gdCorrY =0
gdCorrZ =0
gdCorrRX=0
gdCorrRY=0
gdCorrRZ=0
ENDIF

'-- Reset Measurement movement to finished
gyMeMovFinish=0

'-- WAIT until calculated measurement result is reset
WAIT gxMesResCalc=0
'-- Apply correction to tool data
LP1=P*
LR1=(0,0,0,0,0,0)T
LR1.X=gdCorrX
LR1.Y=gdCorrY
LR1.Z=gdCorrZ
LR1.RX=gdCorrRX
LR1.RY=gdCorrRY
LR1.RZ=gdCorrRZ

S1    MOVE P,P*-LR1,S=10%,A=0,T=giToolNrCorr
LP2=P*
ATDC T=giToolNrCorr,OrgP=LP1,NewP=LP2
DELAY 0.1
```



```
END
'-- Acknowledge error during measurement --
'-- Identify and rectify error cause --
IF gxError=1 THEN
'-- Reset tool data --
Tool[giToolNrCorr]=Tool[giToolNrOrig]
gyReset=1
WAIT gxError=0
gyReset=0
ENDIF

'-- Reset Program Number 1 --
gyPrgNrBit1=0
gyPrgNrBit2=0
gyPrgNrBit3=0
gyPrgNrBit4=0
gyPrgNrBit5=0
END
```




图 4.5 测量工具变形量并执行自动校准

4.7. 工具自动校准完后重新确认作业程序并重启作业

工具校准完成后，确认作业程序并重启原本作业。



图 4.6 确认作业程序位置

如上图所示，可发现在执行现有（原本）的作业程序时，误差被消除。
之后，可继续执行原本作业。

4. 8. 其他作业程序

在示例程序上（用例），由 CALL 指令使用的各种作业程序都是关于参数初始值设置及测量专用圆轨迹生成的内容。

初始值参数设置程序

```
'-- tool number to be corrected
giToolNrCorr=1
giToolNrOrig=25
'-- moving pattern parameter
giStPoseDir=0 '0:Y+, 1:X+, 2:Y-, 3:X-
gdRadius=20
gdMaxTrMov=20 'Max trnaslational moving distance
gdOscDist=30 'center - up,down distance
gdOrientLen=30 'Orientation Length
gdMoveSpeed=100 'Moving speed
END
```

圆轨迹生成程序

```
'-- copy current pose to the start pose --
gpStartP=P*
'-- calculate poses for circle
gpStartP.Z=gpStartP.Z - gdMaxTrMov
gpCirPose1=gpStartP
gpCirPose2=gpStartP
gpCirPose3=gpStartP
gpCirPose4=gpStartP
IF giStPoseDir=0 THEN
  gpCirPose2.X=gpCirPose2.X - gdRadius
  gpCirPose2.Y=gpCirPose2.Y - gdRadius
  gpCirPose3.X=gpCirPose3.X - 0
  gpCirPose3.Y=gpCirPose3.Y - gdRadius*2
  gpCirPose4.X=gpCirPose4.X + gdRadius
  gpCirPose4.Y=gpCirPose4.Y - gdRadius
ELSEIF giStPoseDir=1 THEN
  gpCirPose2.X=gpCirPose2.X - gdRadius
  gpCirPose2.Y=gpCirPose2.Y + gdRadius
  gpCirPose3.X=gpCirPose3.X - gdRadius*2
  gpCirPose3.Y=gpCirPose3.Y + 0
  gpCirPose4.X=gpCirPose4.X - gdRadius
  gpCirPose4.Y=gpCirPose4.Y - gdRadius
ELSEIF giStPoseDir=2 THEN
  gpCirPose2.X=gpCirPose2.X + gdRadius
  gpCirPose2.Y=gpCirPose2.Y + gdRadius
```

```
gpCirPose3.X=gpCirPose3.X + 0  
gpCirPose3.Y=gpCirPose3.Y + gdRadius*2  
gpCirPose4.X=gpCirPose4.X - gdRadius  
gpCirPose4.Y=gpCirPose4.Y + gdRadius  
ELSEIF giStPoseDir=3 THEN  
gpCirPose2.X=gpCirPose2.X + gdRadius  
gpCirPose2.Y=gpCirPose2.Y - gdRadius  
gpCirPose3.X=gpCirPose3.X + gdRadius*2  
gpCirPose3.Y=gpCirPose3.Y - 0  
gpCirPose4.X=gpCirPose4.X + gdRadius  
gpCirPose4.Y=gpCirPose4.Y + gdRadius  
ENDIF  
END
```







● Daegu Office (Head Office)

50, Techno sunhwan-ro 3-gil, yuga, Dalseong-gun, Daegu, 43022, Korea

● GRC

477, Bundangsuseo-ro, Bundang-gu, Seongnam-si, Gyeonggi-do, Korea

● 대구 사무소

(43022) 대구광역시 달성군 유가읍 테크노순환로 3 길 50

● GRC

(13553) 경기도 성남시 분당구 분당수서로 477

● ARS :+82-1588-9997 (A/S center)

● E-mail :robotics@hyundai-robotics.com

